

Het gebruik van Maple bij ConstructieMechanica

MAPLE is een krachtige tool om op een gestructureerde wijze “lastig” en “vervelend” rekenwerk te verrichten. MAPLE 2015 is beschikbaar voor studenten via de campuslicentie van de TU-Delft. Je kunt deze software downloaden via BlackBoard. Belangrijk bij MAPLE is dat je de activation key installeert, deze is 1 jaar geldig.

Om het gebruik van MAPLE te demonstreren worden hier een aantal voorbeelden uitgewerkt.

De onderdelen die aan de orde komen zijn:

- Voorbeeld 1,
- Voorbeeld 2,
- Voorbeeld 3 : het werken met vergelijkingen
- Voorbeeld 4 : het oplossen van een differentiaalvergelijking
- Voorbeeld 5 : werken met matrices

De eerste vier voorbeelden hebben betrekking op statisch onbepaalde constructies. Verschillende oplosmethodieken worden met deze voorbeelden gedemonstreerd.

In voorbeeld 1 wordt gebruik gemaakt van de zakkingslijn voor buiging. Hiervoor worden met behulp van randvoorwaarden en overgangsvoorwaarden acht vergelijkingen opgesteld waarmee het zakkingsveld voor dit voorbeeld kan worden bepaald.


In voorbeeld 2 en 3 wordt met behulp van de krachtenmethode op basis van hoekveranderingsvergelijkingen de krachtsverdelingen bepaald voor twee statisch onbepaalde constructies. Ook hier worden de vergelijkingen opgesteld waarmee de onbekenden kunnen worden opgelost.

In voorbeeld 4 wordt hetzelfde probleem als in voorbeeld 1 in MAPLE direct opgelost door gebruik te maken van de oplossingsmethodieken voor differentiaalvergelijkingen.

In voorbeeld 5 wordt aan de hand van een voorbeeld uit de statica een stelsel vergelijkingen opgesteld dat in matrixvorm kan worden weergegeven. Voor het oplossen van de onbekenden wordt vervolgens gebruik gemaakt van de lineaire algebra technieken die in MAPLE voorhanden zijn.

Voorbeeld 1 : Vergelijkingen oplossen

Het voorbeeld betreft de in figuur 1 weergegeven constructie. De ligger AB wordt belast met een gelijkmatig verdeelde belasting die alleen aangrijpt op het liggerdeel AC. De ligger is in B volledig ingeklemd. Op de ligger werkt een gelijkmatig verdeelde belasting die alleen aangrijpt op liggerdeel AC. De oorsprong van het assenstel x - z -assenstelsel wordt in A gekozen.



Figuur 1 : Statisch onbepaalde ligger

Omdat de belasting niet met één functievoorschrift voor de gehele ligger kan worden beschreven moeten de liggerdelen (velden) AC en CB afzonderlijk worden bekeken¹. Voor elk veld geldt de 4^e orde differentiaalvergelijking voor buiging:

$$EI \frac{d^4 w}{dx^4} = q(x)$$

Om de grootheden w , ϕ , M en V in de velden AC en CB van elkaar te kunnen onderscheiden krijgen ze in veld AC een index I en in veld CB een index II, dus:

$$\begin{array}{ll} \text{voor liggerdeel AC} & : \quad w_I(x), \phi_I(x), \text{ etc.} \\ \text{voor liggerdeel CB} & : \quad w_{II}(x), \text{ etc.} \end{array}$$

De algemene oplossing van de 4^e orde D.V. voor buiging in het geval van een gelijkmatig verdeelde belasting $q(x) = q$ is:

$$w(x) = \underbrace{C_1 + C_2 x + C_3 x^2 + C_4 x^3}_{\text{homogene oplossing}} + \underbrace{\frac{qx^4}{24EI}}_{\text{particuliere oplossing}}$$


Per liggerdeel kan het verplaatsingsveld worden opgesteld. Aangezien de gelijkmatig verdeelde belasting nul is op het liggerdeel BC is de particuliere oplossing voor dit deel van de ligger gelijk aan nul. Er geldt:

$$\begin{aligned} w_I(x) &= C_1 + C_2 x + C_3 x^2 + C_4 x^3 + \frac{qx^4}{24EI} \\ w_{II}(x) &= D_1 + D_2 x + D_3 x^2 + D_4 x^3 \end{aligned} \tag{1}$$

In totaal zijn er 8 onbekende integratieconstanten: C_1 t/m C_4 (veld AC) en D_1 t/m D_4 (veld CB). Deze zijn te berekenen met behulp van de rand- en overgangsvoorwaarden.

¹ Zie Hartsuiker, Toegepaste Mechanica – deel 1, blz. 415, 418 en 429 en deel 2 blz. 29, 504 en 506.

Ter plaatse van de randen A en B gelden (per rand) twee randvoorwaarden. Ter plaatse van de veldovergang in C moeten vier overgangsvoorwaarden worden opgesteld. De rand en overgangsvoorwaarden zijn weergegeven in figuur 2.



Figuur 2 : Rand- en overgangsvoorwaarden

Bij het uitwerken van de rand- en overgangsvoorwaarden wordt gebruik gemaakt van de volgende betrekkingen:

$$\begin{aligned} \varphi_I(x) &= -\frac{dw_I}{dx} & \varphi_{II}(x) &= -\frac{dw_{II}}{dx} \\ M_I(x) &= -EI \frac{d^2 w_I}{dx^2} & \text{en} & \quad M_{II}(x) = -EI \frac{d^2 w_{II}}{dx^2} \\ V_I(x) &= \frac{dM_I}{dx} & V_{II}(x) &= \frac{dM_{II}}{dx} \end{aligned} \quad (2)$$

Samenvatting

De zakkingslijn voor dit buigingsprobleem wordt dus beschreven met de twee functies (1) waarin 8 integratieconstanten voorkomen. Deze 8 onbekenden worden bepaald met behulp van de eerder genoemde 4 randvoorwaarden en 4 overgangsvoorwaarden uit figuur 2.

$$\begin{aligned} w_I(0) &= 0 \\ M_I(0) &= 0 \\ w_I(a) &= w_{II}(a) \\ \varphi_I(a) &= \varphi_{II}(a) \\ M_I(a) &= M_{II}(a) \\ V_I(a) &= V_{II}(a) \\ w_{II}(l) &= 0 \\ \varphi_{II}(l) &= 0 \end{aligned} \quad (3)$$

Bij de uitwerking van de rand- en overgangsvoorwaarden wordt gebruik gemaakt van de onder (2) gegeven betrekkingen.

Uitwerking in MAPLE

In MAPLE kan gewoon met symbolen worden gewerkt. Het is dus niet nodig om alle vergelijkingen eerst met de hand uit te werken. De betrekkingen uit (1), (2) en (3) kunnen direct worden overgenomen in MAPLE. De in rood aangegeven tekst is de in te typen tekst. In blauw is het resultaat dat MAPLE geeft weergegeven. Let er op dat ieder commando met een ; wordt afgesloten.

Na het aanmaken van een nieuw rekenblad wordt gestart met het commando **restart**. Met dit commando worden alle eerder berekende variabelen gewist.

> **restart;**

Vervolgens kunnen de variabelen worden gedeclareerd. Dit zijn voor dit voorbeeld de afmetingen a en l , de belasting q en de buigstijfheid EI . Let erop dat alle grootheden in dezelfde eenheden worden ingevoerd. In dit voorbeeld dus kN en m.

```
> a:=4; l:=10; q:=10; EI:=10000;  
a := 4  
l := 10  
q := 10  
EI := 10000
```

Vervolgens worden de verplaatsingsfuncties (1) voor de beide liggedelen ingevoerd:

```
> w1:=C1+C2*x+C3*x^2+C4*x^3+(1/(24*EI))*q*x^4;  
w1 := C1 + C2 x + C3 x^2 + C4 x^3 +  $\frac{1}{24000} x^4$   
  
> w2:=D1+D2*x+D3*x^2+D4*x^3;  
w2 := D1 + D2 x + D3 x^2 + D4 x^3
```

Voor het verwerken van de randvoorwaarden en overgangsvoorwaarden worden de betrekkingen volgens (2) ingevoerd. Voor het differentiëren in MAPLE wordt gebruik gemaakt van het **diff** commando. Tussen haakjes moet worden aangegeven welke functie gedifferentieerd moet worden en naar welke variabele moet worden gedifferentieerd.

```
> phi1:=-diff(w1,x); phi2:=-diff(w2,x);  
 $\phi1 := -C2 - 2 C3 x - 3 C4 x^2 - \frac{1}{6000} x^3$   
 $\phi2 := -D2 - 2 D3 x - 3 D4 x^2$   
  
> M1:=EI*diff(phi1,x); M2:=EI*diff(phi2,x);  
 $M1 := -20000 C3 - 60000 C4 x - 5 x^2$   
 $M2 := -20000 D3 - 60000 D4 x$   
  
> V1:=diff(M1,x); V2:=diff(M2,x);  
 $V1 := -60000 C4 - 10 x$   
 $V2 := -60000 D4$ 
```

Voor de verdere uitwerking van het probleem moeten nu de 8 vergelijkingen worden ingevoerd. In MAPLE kunnen vergelijkingen expliciet worden vermeld.

Er hoeft dus niets met de hand uitgeschreven of vereenvoudigd te worden. Je kunt zelf een naam verzinnen voor iedere vergelijking. In dit voorbeeld wordt gebruik gemaakt van de afkorting **eq1**, **eq2** etc.

De rand- en overgangsvoorwaarden gelden voor drie verschillende plaatsen in de ligter :

twee randvoorwaarden in A:	$x=0$;
vier overgangsvoorwaarden in C:	$x=a$;
twee randvoorwaarden in B:	$x=l$

Het is van belang dat voor het opstellen van de vergelijkingen aan x de juiste waarde , 0, a of l wordt toegekend en dat na het oplossen van de vergelijkingen x weer als variabele wordt hersteld.

Randvoorwaarden in A ($x=0$)

Bij de twee randvoorwaarden in A moet x de waarde nul krijgen. Uitwerken levert:

```
> x:=0;
                                 $x := 0$ 

> eq1:=w1=0;
                                 $eq1 := C1 = 0$ 

> eq2:=M1=0;
                                 $eq2 := -20000 C3 = 0$ 
```

Overgangsvoorwaarden in C ($x=a$)

De vier overgangsvoorwaarden worden als volgt ingevoerd:

```
> x:=a;
                                 $x := 4$ 

> eq3:=w1=w2;
                                 $eq3 := C1 + 4 C2 + 16 C3 + 64 C4 + \frac{4}{375} = D1 + 4 D2 + 16 D3 + 64 D4$ 

> eq4:=phi1=phi2;
                                 $eq4 := -C2 - 8 C3 - 48 C4 - \frac{4}{375} = -D2 - 8 D3 - 48 D4$ 

> eq5:=M1=M2;
                                 $eq5 := -20000 C3 - 240000 C4 - 80 = -20000 D3 - 240000 D4$ 

> eq6:=V1=V2;
                                 $eq6 := -60000 C4 - 40 = -60000 D4$ 
```

Randvoorwaarden in B ($x=l$)

De twee randvoorwaarden in B leiden tot:

```
> x:=l;
                                 $x := 10$ 

> eq7:=w2=0;
                                 $eq7 := D1 + 10 D2 + 100 D3 + 1000 D4 = 0$ 

> eq8:=phi2=0;
                                 $eq8 := -D2 - 20 D3 - 300 D4 = 0$ 
```

Alle informatie is nu ingevoerd. De 8 vergelijkingen met 8 onbekenden moeten nu alleen nog worden opgelost. Hiervoor beschikt MAPLE over een “solver” die wordt aangeroepen met **solve**. De oplossing wordt eerst opgeslagen in een variabele waarvan de naam vrij mag worden gekozen. In dit geval wordt de naam **solution** gebruikt.

```
> solution:=solve({eq1,eq2,eq3,eq4,eq5,eq6,eq7,eq8},
                  {C1,C2,C3,C4,D1,D2,D3,D4});
solution := { C1 = 0, C3 = 0, D4 =  $\frac{73}{375000}$ , C2 =  $\frac{41}{3750}$ , D1 =  $\frac{-4}{375}$ , D3 =  $\frac{-1}{250}$ , C4 =  $\frac{-59}{125000}$ ,
              D2 =  $\frac{27}{1250}$  }
```

Vervolgens wordt met het commando **assign** de oplossing toegewezen aan de 8 variabelen C₁ t/m D₄.

```
> assign(solution);
```

Het probleem is nu opgelost. Om de oplossing te kunnen tekenen moet de waarde van x , die na het formuleren van de randvoorwaarden in B nog op l staat, worden gewist. Immers x moet weer een “echte” variabele worden.

```
> x:='x';
x := x
```

De gevonden verplaatsingsvelden kunnen netjes worden weergegeven met:

```
> print(w1); print(w2);
```

$$\frac{41}{3750}x - \frac{59}{125000}x^3 + \frac{1}{24000}x^4$$


$$-\frac{4}{375} + \frac{27}{1250}x - \frac{1}{250}x^2 + \frac{73}{375000}x^3$$

Dit resultaat kan ook grafisch worden weergegeven. Om in één figuur de totale oplossing te kunnen tekenen moeten de “plotjes” van beide zakkingslijnen w_I en w_{II} worden gecombineerd. Hieronder is weergegeven hoe dat in MAPLE wordt ingevoerd.

```
> with(plots):
F:=plot(w1,x=0..a,y=-0.05..0.05,
        labels=["x-as","w"],title="Zakking",style=line):
G:=plot(w2,x=a..l,y=-0.05..0.05,style=line):
display({F,G});
```

Met dit commando krijgen de plaatjes voor w_I en w_{II} eerst een eigen tijdelijke naam F en G waarna met het **display** commando deze plaatjes in één figuur worden afgebeeld. Het resultaat is hiernaast weergegeven. Merk op dat MAPLE de positieve as naar boven uitzet.


Uiteraard kunnen op soortgelijke wijze ook de figuren voor de dwarskracht, het moment en de hoekverdraaiing worden samengesteld. Dit wordt aan de lezer overgelaten.



Figuur 3 : Zakking

Voorbeeld 2 : Ongeschoord raamwerk

Het onderstaande portaal is een ongeschoord raamwerk dat zowel horizontaal als verticaal wordt belast. De buigstijfheid van de regel is EI , die van de kolommen is mEI en nEI , zoals aangegeven in figuur 4.




Figuur 4 : Ongeschoord raamwerk

Voor deze constructie zal de krachtsverdeling worden bepaald m.b.v. de methode van hoekveranderingsvergelijkingen. (zie collegedictaat CT2031 van Hartsuijker en Welleman).

Analyse:

Dit raamwerk is een raamwerk met verplaatsbare knopen. Als de momentvaste verbindingen in C en D door scharnieren worden vervangen met de daarbij behorende onbekende momentenparen M_C en M_D ontstaat een mechanisme met één vrijheidsgraad waarvoor de rotatie θ van de kolommen wordt gekozen. Deze aanpak leidt dus tot de 3 onbekenden M_C , M_D en θ , zie figuur 5.



Figuur 5 : De Statisch onbepaalde M_C , M_D en de onbekende vrijheidsgraad θ

Er zijn drie vergelijkingen nodig voor het bepalen van deze onbekenden :

- twee vormveranderingsvoorwaarden in de vorm van hoekveranderingsvergelijkingen
- één evenwichtsvergelijking in de vorm van virtuele arbeid

Het uitwerken van de hoekveranderingsvergelijkingen kan vaak m.b.v. de *vergeet-mij-nietjes* voor liggers op twee steunpunten. De positieve rotaties van de staafuiteinden worden in dezelfde richting gekozen als de aangenomen momenten. Het opstellen van de vergelijkingen levert zodoende:

$$(1) \quad \theta_C^{CA} + \theta_C^{CD} = 0 \quad \Rightarrow \quad \left(\frac{M_C h}{3nEI} - \theta \right) + \left(\frac{M_C l}{3EI} + \frac{\beta F l^2}{16EI} - \frac{M_D l}{6EI} \right) = 0$$

$$(2) \quad \theta_D^{DC} + \theta_D^{DB} = 0 \quad \Rightarrow \quad \left(-\frac{M_C l}{6EI} - \frac{\beta F l^2}{16EI} + \frac{M_D l}{3EI} \right) + \left(\frac{M_D h}{3mEI} - \theta \right) = 0$$

$$(3) \quad \delta A = 0 \quad \Rightarrow \quad \alpha F \times \delta \theta \times h - M_C \times \delta \theta - M_D \times \delta \theta = 0 \quad \Leftrightarrow \quad M_C + M_D = \alpha F h$$

De horizontale verplaatsing u in C is gelijk aan die in D en is groot $\theta \times h$.

Oplossingsfase:

De drie vergelijkingen kunnen met de hand maar bij voorkeur met MAPLE worden opgelost.

TIP : Strategie bij het oplossen met de hand is door uit de eerste twee vergelijkingen de θ te elimineren en vervolgens dit resultaat te combineren met vergelijking (3).

Uitwerken met MAPLE levert:

```

> restart;
> eq1:=(Mc*h/(3*n*EI))-theta+(Mc*L/(3*EI))+((beta*F*L^2)/(16*EI))-((Md*L)/(6*EI))=0;
      eq1 := \frac{Mc h}{3 n EI} - \theta + \frac{Mc L}{3 EI} + \frac{\beta F L^2}{16 EI} - \frac{Md L}{6 EI} = 0
> eq2:=(-Mc*L/(6*EI))-((beta*F*L^2)/(16*EI))+((Md*L)/(3*EI))+((Md*h)/(3*m*EI))-theta=0;
      eq2 := -\frac{Mc L}{6 EI} - \frac{\beta F L^2}{16 EI} + \frac{Md L}{3 EI} + \frac{Md h}{3 m EI} - \theta = 0
> eq3:=Mc+Md=alpha*F*h;
      eq3 := Mc + Md = \alpha F h
> solve({eq1,eq2,eq3},{Mc,Md,theta}):
> assign(%);
> Mc:Md:theta:
> Mc;
      \frac{F n (-3 \beta L^2 m + 12 L m \alpha h + 8 \alpha h^2)}{8 (m h + 3 m L n + n h)}
> Md;
      \frac{m F (3 \beta L^2 n + 8 \alpha h^2 + 12 L n \alpha h)}{8 (m h + 3 m L n + n h)}
> theta;
      \frac{F h (16 L m \alpha h + 12 L^2 m n \alpha + 3 \beta L^2 n + 16 \alpha h^2 + 16 L n \alpha h - 3 \beta L^2 m)}{48 EI (m h + 3 m L n + n h)}
  
```

De drie onbekenden zijn daarmee opgelost. Uit de bovenstaande MAPLE afbeelding kan worden afgelezen dat hiervoor de volgende uitdrukkingen gelden:

$$M_C = \frac{12mnl + 8nh}{8(m+n)h + 24mnl} \alpha Fh - \frac{3mnl}{8(m+n)h + 24mnl} \beta Fl$$

$$M_D = \frac{12mnl + 8mh}{8(m+n)h + 24mnl} \alpha Fh + \frac{3mnl}{8(m+n)h + 24mnl} \beta Fl$$

$$\theta = \frac{16h(ml + nl + h) + 12mnl^2}{(8(m+n)h + 24mnl)} \frac{\alpha Fh}{6EI} + \frac{3(n-m)hl}{(8(m+n)h + 24mnl)} \frac{\beta Fl}{6EI} \quad \text{en} \quad u_C = u_D = \theta \times h$$

Positieve waarden voor de hierboven gevonden momenten komen overeen met de aangenomen richtingen zoals weergegeven in figuur 5 op de vorige bladzijde.


Voor het kunnen tekenen van de M -lijn is het van belang ook het moment in E te weten. Ga zelf na dat met de bekende momenten in C en D en de momentensom hiervoor geldt:

$$M_E = \frac{M_C - M_D}{2} + \frac{1}{4} \beta Fl = \frac{4h(n-m)}{8(m+n)h + 24mnl} \alpha Fh + \frac{(3mnl + 2mh + 2nh)}{8(m+n)h + 24mnl} \beta Fl$$

De uitdrukkingen zijn gesplitst in de bijdragen van de afzonderlijke krachten αF en βF .

Interpretatie van de resultaten

Voor een gegeven situatie kan de M -lijn worden getekend. Belangrijk daarbij zijn de afspraken voor positieve krachtsgrootheden. Deze zijn in figuur 6 weergegeven.



Figuur 6 : Positieve richtingen voor de krachtsgrootheden

De oplegreacties kunnen nu worden bepaald:

$$A_H = \frac{M_C}{h} = \frac{12mnl + 8nh}{8(m+n)h + 24mnl} \alpha F - \frac{3mnl}{8(m+n)h + 24mnl} \times \frac{l}{h} \beta F$$

$$A_V = -\frac{h}{l} \alpha F + \frac{1}{2} \beta F$$

$$B_H = \frac{M_D}{h} = \frac{12mnl + 8mh}{8(m+n)h + 24mnl} \alpha F + \frac{3mnl}{8(m+n)h + 24mnl} \times \frac{l}{h} \beta F$$

$$B_V = \frac{h}{l} \alpha F + \frac{1}{2} \beta F$$

In het bijzondere geval dat de kolommen dezelfde buigstijfheid hebben geldt $m=n$ en gaan de bovenstaande vergelijkingen over in:

$$A_H = \frac{M_C}{h} = \frac{1}{2} \alpha F - \frac{3nl}{16h + 24nl} \times \frac{l}{h} \beta F$$

$$A_V = -\frac{h}{l} \alpha F + \frac{1}{2} \beta F$$

$$B_H = \frac{M_D}{h} = \frac{1}{2} \alpha F + \frac{3nl}{16h + 24nl} \times \frac{l}{h} \beta F$$


$$B_V = \frac{h}{l} \alpha F + \frac{1}{2} \beta F$$

Opmerkingen:

- Hieruit blijkt dat indien een constructie met identieke kolommen alleen belast wordt met een horizontale belasting, de oplegreacties onafhankelijk zijn van de buigstijfheden van zowel de regel als de kolom. Merk op dat dit niet geldt in het algemene geval wanneer de kolommen verschillend worden uitgevoerd.
- Als de kolommen gelijke buigstijfheden hebben ($m=n$) blijkt het moment in E onafhankelijk te zijn van de grootte van de horizontale kracht αF .

Voorbeeld 3 : Ongeschoord raamwerk met veren

Het onderstaande portaal is een ongeschoord raamwerk dat zowel horizontaal als verticaal wordt belast. De buigstijfheid van de regel is EI , die van de kolommen is nEI , zoals aangegeven in figuur 7. De staven zijn verend met elkaar verbonden m.b.v. lineair elastische (LE) rotatieveren die een veerstijfheid k hebben.




Figuur 7 : Ongeschoord raamwerk

Voor deze constructie zal de krachtsverdeling worden bepaald m.b.v. de methode van hoekveranderingsvergelijkingen. (zie collegedictaat CT2031 van Hartsuijker en Welleman).

Analyse:

Het voorbeeld vertoont grote gelijkenis met het voorgaande voorbeeld. Nieuw element in het geheel is de verende verbinding tussen de staven. Hierdoor ontstaat een extra vervorming. Dit is de hoekverdraaiing in de rotatieveer. Deze vervorming wordt aangeduid met $\Delta\theta$. Uiteraard is deze vervorming recht evenredig met het moment in de veer aangezien een LE-veer karakteristiek wordt aangenomen. Ook nu leidt deze aanpak dus tot de 3 onbekenden M_C , M_D en θ , zie figuur 8.



Figuur 8 : De Statisch onbepaalde M_C , M_D en de onbekende vrijheidsgraad θ

De drie benodigde vergelijkingen zijn voor het oplossen van de onbekenden zijn:

- twee vormveranderingsvoorwaarden in de vorm van hoekveranderingsvergelijkingen
- één evenwichtsvergelijking in de vorm van virtuele arbeid

De positieve rotaties van de staafuiteinden worden in dezelfde richting gekozen als de aangenomen momenten. Het opstellen van de vergelijkingen levert zodoende:

$$(1) \quad \theta_C^{CA} + \theta_C^{CD} + \Delta\theta_C = 0 \quad \Rightarrow \quad \left(\frac{M_C h}{3nEI} - \theta \right) + \left(\frac{M_C l}{3EI} + \frac{\beta F l^2}{16EI} - \frac{M_D l}{6EI} \right) + \frac{M_C}{k} = 0$$

$$(2) \quad \theta_D^{DC} + \theta_D^{DB} + \Delta\theta_D = 0 \quad \Rightarrow \quad \left(-\frac{M_C l}{6EI} - \frac{\beta F l^2}{16EI} + \frac{M_D l}{3EI} \right) + \left(\frac{M_D h}{3nEI} - \theta \right) + \frac{M_D}{k} = 0$$


$$(3) \quad \delta A = 0 \quad \Rightarrow \quad \alpha F \times \delta\theta \times h - M_C \times \delta\theta - M_D \times \delta\theta = 0 \quad \Leftrightarrow \quad M_C + M_D = \alpha F h$$

De horizontale verplaatsing u in C is gelijk aan die in D en is groot $\theta \times h$.

Oplossingsfase:

De drie vergelijkingen kunnen met de hand maar bij voorkeur met MAPLE worden opgelost.

Op een handige manier uitwerken met MAPLE levert:



```

> restart;
> eq1:=(Mc*h/(3*n*EI))-theta+(Mc*L/(3*EI))+((F*L^2)/(16*EI))-((Md*L)/(6*EI))+(Mc/k)=0;
      eq1 = \frac{Mc h}{3 n EI} - \theta + \frac{Mc L}{3 EI} + \frac{F L^2}{16 EI} - \frac{Md L}{6 EI} + \frac{Mc}{k} = 0
> eq2:=(-Mc*L/(6*EI))-((F*L^2)/(16*EI))+((Md*L)/(3*EI))+(Md*h/(3*n*EI))-theta+(Md/k)=0;
      eq2 = -\frac{Mc L}{6 EI} - \frac{F L^2}{16 EI} + \frac{Md L}{3 EI} + \frac{Md h}{3 n EI} - \theta + \frac{Md}{k} = 0
> eq3:=Mc+Md=alpha*F*h;
      eq3 = Mc + Md = \alpha F h
> solve({eq1,eq2,eq3},{Mc,Md,theta}):
> assign(%);
> Mc:Md:theta:
> expand(Mc,alpha*F*h);
      -\frac{3 F L^2 n k}{8 (2 h k + 3 L n k + 6 n EI)} + \frac{F h^2 k \alpha}{2 h k + 3 L n k + 6 n EI} + \frac{3 F n EI \alpha h}{2 h k + 3 L n k + 6 n EI} + \frac{3 F L n k \alpha h}{2 (2 h k + 3 L n k + 6 n EI)}
> expand(Md,alpha*F*h);
      \frac{3 F L^2 n k}{8 (2 h k + 3 L n k + 6 n EI)} + \frac{F h^2 k \alpha}{2 h k + 3 L n k + 6 n EI} + \frac{3 F n EI \alpha h}{2 h k + 3 L n k + 6 n EI} + \frac{3 F L n k \alpha h}{2 (2 h k + 3 L n k + 6 n EI)}
> expand(theta,alpha*F*h);
      \frac{\alpha F h^2}{6 n EI} + \frac{\alpha F h}{2 k} + \frac{\alpha F h L}{12 EI}
> u:=expand(theta*h,alpha*F*h^2);
      u = \frac{\alpha F h^3}{6 n EI} + \frac{\alpha F h^2}{2 k} + \frac{\alpha F h^2 L}{12 EI}

```

De drie onbekenden zijn daarmee opgelost en tevens al gesplitst in de diverse aandelen. Uit de bovenstaande afbeelding kunnen, door hergroeperen, de onderstaande uitdrukkingen worden verkregen:

$$M_C = \frac{1}{2} \alpha F h - \frac{1}{3 + 2 \frac{h}{nl} + \frac{6EI}{kl}} \frac{3}{8} Fl$$

$$M_D = \frac{1}{2} \alpha F h + \frac{1}{3 + 2 \frac{h}{nl} + \frac{6EI}{kl}} \frac{3}{8} Fl$$

$$\theta = \left(\frac{h}{6nEI} + \frac{l}{12EI} + \frac{1}{2k} \right) \alpha F h \quad \text{en} \quad u_C = u_D = \theta \times h = \frac{1}{2} \alpha F h^2 \left(\frac{h}{3nEI} + \frac{l}{6EI} + \frac{1}{k} \right)$$

Positieve waarden komen overeen met de richtingen die werden aangenomen in figuur 8 op de vorige bladzijde.


De oplegreacties kunnen op dezelfde wijze als in het voorgaande voorbeeld worden bepaald.

Interpretatie van de resultaten

De horizontale verplaatsing van de bovenregel blijkt onafhankelijk te zijn van de grootte van de verticale belasting. Dit was in het vorige voorbeeld ook het geval voor de bijzondere situatie dat beide kolommen gelijk worden uitgevoerd en dat geldt dus ook in dit geval.

Als de uitdrukking voor de horizontale verplaatsing beter bekeken wordt zijn daar drie afzonderlijke aandelen in te herkennen:

$$u_c = \frac{1}{2} \alpha F h^2 \left(\frac{h}{3nEI} + \frac{l}{6EI} + \frac{1}{k} \right)$$




Voorbeeld 4 : Oplossen van een Differentiaal Vergelijking (DV)

In MAPLE is het ook mogelijk differentiaalvergelijkingen op te lossen. Met behulp van het voorbeeld 1 zal dit worden gedemonstreerd. Het betreft hier de 4^e orde D.V. voor buiging waarvoor geldt:

$$EI \frac{d^4 w}{dx^4} = -q(x)$$

De ligger AB wordt belast met een gelijkmatig verdeelde belasting die alleen aangrijpt op het liggerdeel AC. De ligger is in B volledig ingeklemd. Op de ligger werkt een gelijkmatig verdeelde belasting die alleen aangrijpt op liggerdeel AC. De oorsprong van het assenstel x - z -assenstelsel wordt in A gekozen.




Figuur 9 : Statisch onbepaalde ligger

In voorbeeld 1 werd deze constructie opgesplitst in twee delen aangezien de belasting niet met één functievoorschrift voor de gehele ligger kan worden beschreven en werd de met de hand bepaalde algemene oplossing van de DV gebruikt. Deze stappen kunnen echter ook direct met MAPLE worden uitgevoerd. Door de belasting met een stapfunctie (**Heaviside** functie) in te voeren is het mogelijk deze met één functievoorschrift voor het liggerdeel AB in MAPLE te beschrijven.

```
> q:=Q*(1-Heaviside(x-a));
      q := 10 - 10 Heaviside(x - 4.0)

> plot(q, x=0..L);
```



De D.V. kan in MAPLE worden ingevoerd als een vergelijking:

```
> DV:=EI*diff(w(x), x$4)=q;
      DV := 10000. (d^4 w(x) / dx^4) = 10 - 10 Heaviside(x - 4.0)

      ( met x$4 wordt de vierde afgeleide naar x bedoeld )
```

Naast de D.V. moeten uiteraard ook geldige randvoorwaarden worden gespecificeerd.

Voor de oplegging in A geldt : $w(0)=0$ en $M(0)=0$

Voor de inklemming in B geldt : $w(L)=0$ en $\phi = 0$

```
> RV:=w(0)=0, (D@@2)(w)(0)=0, w(L)=0, D(w)(L)=0;
```


In deze set van randvoorwaarden wordt met $(D@@2)(w)(0)$ de tweede afgeleide van de zakkingsfunctie $w(x)$ bedoeld voor $x=0$. Met $D(w)(L)$ wordt de eerste afgeleide van $w(x)$ voor $x=L$ bedoeld.

Met het invoeren van de D.V. en het aangeven van de r.v.w. kan MAPLE de oplossing direct bepalen met behulp van het commando **dsolve**.

```
> dsolve({DV, RV}, {w(x)}): w:=rhs(%):
```

In dit commando zijn DV en RV de bekenden, de onbekende is de zakkingsfunctie $w(x)$. Het commando **dsolve** levert een oplossing die helaas nog niet is toegekend aan de variabele w . In het tweede commando hierboven na de “:” wordt het rechterlid (**rhs**=right hand side) van de oplossing, aangegeven met een %-teken toegekend aan w .

Het zakkingsveld is nu bepaald. In de uitdrukking voor de zakking komen ook weer termen voor met heaviside-functies waardoor het niet zinvol is deze hier af te beelden. Uiteraard is een plot van deze functie wel zinvol. Op basis van het gevonden verplaatsingsveld kunnen ook de M - en V -lijnen worden weergegeven. De gevonden blauwe zakkingslijn in figuur 10 komt overeen met de eerder bepaalde oplossing van voorbeeld 1.



Figuur 10 : Resultaten voor gegeven parameters

Naast de stapfunctie die hier gebruikt is voor een gelijkmatig verdeelde belasting die op een deel van de ligter aangrijpt kan er ook gebruik worden gemaakt van een Dirac functie om een puntlast in te voeren. Werk dit zelf maar eens uit.


Opmerking:

Randvoorwaarden kunnen ook als losse vergelijkingen worden ingevoerd en opgelost nadat eerst de algemene oplossing met **dsolve()** is verkregen. Voorbeelden hiervan zijn te vinden in het bestand `maple.zip` dat op de site kan worden gedownload.

Voorbeeld 5 : Werken met matrices

Het tweede voorbeeld betreft het evenwicht van een star blok zoals in deel 1 van Hartsuijker op pagina 89 wordt beschreven.

De kubus met riblengte a en gewicht G wordt in evenwicht gehouden door zes krachten F_1 t/m F_6 . Gegeven is dat voor de hoek α tussen de werklijnen van de krachten geldt: $\tan \alpha = 3/4$.



Figuur 11 : Kubus met krachten

De evenwichtsvergelijkingen die voor dit probleem kunnen worden opgesteld zijn:

$$\begin{aligned}\sum F_x &= \frac{3}{5} F_2 + \frac{3}{5} F_6 = 0 \\ \sum F_y &= F_1 + \frac{4}{5} F_2 + F_3 + \frac{4}{5} F_4 + F_5 + \frac{4}{5} F_6 = 0 \\ \sum F_z &= \frac{3}{5} F_4 - G = 0 \\ \sum T_x &= \left(F_1 + \frac{4}{5} F_2 + F_3 \right) \times a - G \times \frac{1}{2} a = 0 \\ \sum T_y &= \frac{3}{5} F_2 \times a - \frac{3}{5} F_4 \times a + G \times \frac{1}{2} a = 0 \\ \sum T_z &= - \left(F_3 + \frac{4}{5} F_4 + F_5 \right) \times a = 0\end{aligned}$$

Dit stelsel vergelijkingen kan in matrix-vorm worden weergegeven:

$$\begin{bmatrix} 0 & \frac{3}{5} & 0 & 0 & 0 & \frac{3}{5} \\ 1 & \frac{4}{5} & 1 & \frac{4}{5} & 1 & \frac{4}{5} \\ 0 & 0 & 0 & \frac{3}{5} & 0 & 0 \\ a & \frac{4}{5} a & a & 0 & 0 & 0 \\ 0 & \frac{3}{5} a & 0 & -\frac{3}{5} a & 0 & 0 \\ 0 & 0 & -a & -\frac{4}{5} a & a & 0 \end{bmatrix} \times \begin{bmatrix} F_1 \\ F_2 \\ F_3 \\ F_4 \\ F_5 \\ F_6 \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ G \\ \frac{1}{2} Ga \\ -\frac{1}{2} Ga \\ 0 \end{bmatrix}$$

De vector met onbekenden wordt gevormd door de krachten F_i . Dit stelsel kan formeel geschreven worden als:

$$[A]\{x\} = \{b\}$$

Uitwerking in MAPLE

Allereerst wordt een schoon MAPLE werkblad gemaakt en worden eventuele waarden van de variabelen gewist:

```
> restart;
```

```
> G:=24; a:=1;
```

```
G := 24
```

```
a := 1
```

Het oplossen van matrices gebeurt met behulp van de bibliotheek (library) **linalg**. Deze moet aangeroepen worden om de oplosroutines actief te maken. Het commando daarvoor is:

```
> with(linalg):
```

```
Warning, the protected names norm and trace have been redefined and  
unprotected
```

De waarschuwing die MAPLE geeft is niet relevant.

Vervolgens kunnen de matrix en de vector met bekenden worden ingevoerd. Je mag zelf een naam geven aan zowel de matrix als de vector met bekenden. In dit voorbeeld wordt de matrix **A** en de vector **b** genoemd. De invoer moet zorgvuldig gebeuren dus let op de syntax :

```
>
```

```
A:=matrix([[0, (3/5), 0, 0, 0, (3/5)], [1, (4/5), 1, (4/5), 1, (4/5)], [0,  
0, 0, (3/5), 0, 0], [a, (4/5)*a, a, 0, 0, 0], [0, (3/5)*a, 0, (-  
3/5)*a, 0, 0], [0, 0, -a, (-4/5)*a, -a, 0]]);
```

$$A := \begin{bmatrix} 0 & \frac{3}{5} & 0 & 0 & 0 & \frac{3}{5} \\ 1 & \frac{4}{5} & 1 & \frac{4}{5} & 1 & \frac{4}{5} \\ 0 & 0 & 0 & \frac{3}{5} & 0 & 0 \\ 1 & \frac{4}{5} & 1 & 0 & 0 & 0 \\ 0 & \frac{3}{5} & 0 & -\frac{3}{5} & 0 & 0 \\ 0 & 0 & -1 & -\frac{4}{5} & -1 & 0 \end{bmatrix}$$

```
> b:=vector([0, 0, G, 0.5*G*a, -0.5*G*a, 0]);
```

```
b := [0, 0, 24, 12.0, -12.0, 0]
```

Het oplossen van de vector met onbekenden, de zes krachten gaat als volgt:

```
> x:=linsolve(A,b);
```

```
x := [-0., 20.00000002, -4.00000002, 40.00000001, -27.99999999, -20.00000002]
```

Vergelijk zelf deze uitkomst met die uit het boek.

Opmerkingen t.a.v. Maple

- Opdrachten in Maple moeten worden afgesloten door een ';' of een ':'. Als na een ';' de Enter-toets wordt ingedrukt, wordt de opdracht ingevoerd en de reactie van Maple verschijnt op het scherm. Als je na een ':' de Enter-toets indrukt, wordt de opdracht ook ingevoerd en door Maple verwerkt. Er verschijnt echter geen reactie op het scherm.
- Het laatst berekende resultaat kan met een % worden opgehaald. De voorlaatste wordt aangeduid met %% en het daarvoor bepaalde resultaat met %%%. Dit is vaak een snelle syntax hoewel het wel ten koste gaat van de leesbaarheid van het werkblad.
- In MAPLE zit een uitgebreide helpfunctie. Met de functietoets F1 kan snel help worden opgevraagd. Zoeken in de help is in het begin even wennen. Meestal staan onderaan de help-pagina's handige voorbeelden die met copy en paste snel uitgeprobeerd kunnen worden.
- Voor het invoeren van nieuwe MAPLE commando's tussen bestaande regels moet in het menu gebruik worden gemaakt van:

Insert -> Execution Group -> Before cursor

- Voor het verwijderen van een commando moet in het menu gebruik worden gemaakt van:

Edit -> Delete paragraph

- Let er op dat bij fouten altijd het rekenblad met de menu-handeling:

Edit -> Execute -> Worksheet

opnieuw wordt doorgerekend. Anders bestaat de kans dat variabelen toch "oude" niet bedoelde waarden behouden.

Hiermee zijn de meest voorkomende MAPLE handelingen uitgelegd en mag het gebruik van MAPLE, voor de bij ConstructieMechanica voorkomende opgaven, geen problemen meer opleveren.